
シリーズ No.5　PIC 入門アセンブラ編

解答例集

解答例をあげておきます。
あくまで例ですから，この通りでなければならないことはありません。
特にプログラムの正解は１つではありません。解答例を参考に皆さんで工夫してみてください。

シリーズ No.5　PIC 入門アセンブラ編

05-1- 解答例

解答例集

配線する線材の色は自分でルールを決めておきましょう。
不具合が起きた時，原因を究明しやすくなります。

04- 解答例

シリーズ No.5　PIC 入門アセンブラ編

07-1- 解答例

05-2- 解答例

解答例集

B'00001111'

07-1- 解答例とは，LED と抵抗の並びが逆ですが（PIC →抵抗
→ LED と PIC → LED →抵抗）この場合どちらでも構いません。



4

シリーズ No.5　PIC 入門アセンブラ編

10-1- 解答例

解答例集

STEP09.asm09- 解答例

 LIST      P=PIC16F84A
 INCLUDE   P16F84A.INC

 __CONFIG _HS_OSC & _WDT_OFF & _PWRTE_ON & _CP_OFF

 BSF	 STATUS, RP0	 ; BANK 1 に切替え
 CLRF	 TRISA		  ; A ポートを全て出力に設定
 CLRF	 TRISB		  ; B ポートを全て出力に設定
 BCF	 STATUS, RP0	 ; BANK 0 に切替え

 MOVLW 	B'00001111'	 ;リテラル → W レジスタ
 MOVWF	 PORTA		  ; W レジスタ → ファイルレジスタ　PORTA に出力

LOOP
 GOTO	 LOOP
 END

解答例集

10-2- 解答例

STEP14-3.asm14-3- 解答例

 list      p=16F84A

 #include <p16F84A.inc>

 __CONFIG _HS_OSC & _WDT_OFF & _PWRTE_ON & _CP_OFF

	 ORG 00h			  ; 00h 番地からプログラムをセット

	 BSF	 STATUS, RP0	 ; BANK 1 に切替え

	 MOVLW 	 B'00111'	 ; RA0,RA1,RA2 を入力に設定 → W レジスタ

	 MOVWF	 TRISA		  ; W レジスタ → A ポート 入出力設定

	 CLRF	 TRISB		  ; B ポートを全て出力に設定

	 BCF	 STATUS, RP0	 ; BANK 0 に切替え

;******************************

; 　　　　メインリスト

;******************************

MAIN

	 BTFSS	 PORTA,0		 ; RA0 が 1 なら次の行をスキップ

	 GOTO	 LED_ON4		 ; (SW0 ON)

	 BTFSS	 PORTA,1		 ; RA1 が 1 なら次の行をスキップ

	 GOTO	 LED_ON3		 ; (SW0 OFF) (SW1 ON)

	 BTFSS	 PORTA,2		 ; RA2 が 1 なら次の行をスキップ

	 GOTO	 LED_ON2		 ; (SW0 OFF) (SW1 OFF) (SW2 ON)

	 GOTO	 LED_OFF		 ; (SW0 OFF) (SW1 OFF) (SW2 OFF)

;******************************

;	 LED 点灯処理 4 灯 (SW0 ON)

;******************************

LED_ON4

	 MOVLW	 B'00001111'	 ; RB0~RB3 点灯パターン → W レジスタ

	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力

	 GOTO	 MAIN		  ; MAIN へジャンプ

「14-2.asm」に赤文字部分を追加，変更

解答例集

;******************************

;	 LED 点灯処理 3 灯 (SW1 ON)

;******************************

LED_ON3

	 MOVLW	 B'00000111'	 ; RB0~RB2 点灯パターン → W レジスタ

	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力

	 GOTO	 MAIN		  ; MAIN へジャンプ

;******************************

;	 LED 点灯処理 2 灯 (SW2 ON)

;******************************

LED_ON2

	 MOVLW	 B'00000011'	 ; RB0~RB1 点灯パターン → W レジスタ

	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力

	 GOTO	 MAIN		  ; MAIN へジャンプ

;******************************

;	 LED 消灯処理 (SW0 OFF) (SW1 OFF) (SW2 OFF)

;******************************

LED_OFF

	 MOVLW	 B'00000000'	 ; 全消灯パターン → W レジスタ

	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力

	 GOTO	 MAIN		  ; MAIN へジャンプ

END

解答例集

STEP14-4.asm14-4- 解答例

 list      p=16F84A

 #include <p16F84A.inc>

 __CONFIG _HS_OSC & _WDT_OFF & _PWRTE_ON & _CP_OFF

	 ORG 00h			  ; 00h 番地からプログラムをセット

	 BSF	 STATUS, RP0	 ; BANK 1 に切替え

	 MOVLW 	 B'00011'	 ; RA0,RA1 を入力に設定 → W レジスタ

	 MOVWF	 TRISA		  ; W レジスタ → A ポート 入出力設定

	 CLRF	 TRISB		  ; B ポートを全て出力に設定

	 BCF	 STATUS, RP0	 ; BANK 0 に切替え

;******************************

; 　　　　メインリスト

;******************************

MAIN

	 BTFSC	 PORTA,0		 ; RA0 が 0 なら次の行をスキップ

	 GOTO	 SW0_OFF		 ; (SW0 OFF)

	 GOTO	 LED_ON4		 ; (SW0 ON)

SW0_OFF

	 BTFSC	 PORTA,1		 ; RA1 が 0 なら次の行をスキップ

	 GOTO	 LED_OFF		 ; (SW0 OFF) (SW1 OFF)

	 GOTO	 LED_ON3		 ; (SW0 OFF) (SW1 ON)

;******************************

;	 LED 点灯処理 4 灯 (SW0 ON) (SW1 OFF)

;******************************

LED_ON4

	 MOVLW	 B'00001111'	 ; RB0~RB3 点灯パターン → W レジスタ

	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力

	 GOTO	 MAIN		  ; MAIN へジャンプ

メインリストが違います。つなげて書くことができないので，ラベル「SW0_OFF」を付けてジャンプしています。

解答例集

;******************************

;	 LED 点灯処理 3 灯 (SW0 OFF) (SW1 ON)

;******************************

LED_ON3

	 MOVLW	 B'00000111'	 ; RB0~RB2 点灯パターン → W レジスタ

	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力

	 GOTO	 MAIN		  ; MAIN へジャンプ

;******************************

;	 LED 消灯処理 (SW0 OFF) (SW1 OFF)

;******************************

LED_OFF

	 MOVLW	 B'00000000'	 ; 全消灯パターン → W レジスタ

	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力

	 GOTO	 MAIN		  ; MAIN へジャンプ

END

解答例集

シリーズ No.5　PIC 入門アセンブラ編

解答例集

15-1- 解答例

シリーズ No.5　PIC 入門アセンブラ編

17-1- 解答例

解答例集

19-1- 解答例

STEP19-2.asm19-2- 解答例

〜 省略 〜　初期設定

;******************************
; 	  レジスタ設定
;******************************

	 T1	 EQU	 0CH
	 T2	 EQU	 0DH
	 T3	 EQU	 0EH

	 BSF	 STATUS, RP0	 ; BANK 1 に切替え
	 CLRF	 TRISA		  ; A ポートを全て出力に設定
	 CLRF	 TRISB		  ; B ポートを全て出力に設定
	 BCF	 STATUS, RP0	 ; BANK 0 に切替え

;******************************
; 　　　　メインリスト
;******************************

MAIN				    ; ラベル
	 MOVLW 	 B'00000001'	 ; ●●●○　リテラル → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ　PORTB に出力

	 CALL	 TIM500		  ; 0.5 秒タイマ

	 MOVLW 	 B'00000010'	 ; ●●○●　リテラル → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ　PORTB に出力

	 CALL	 TIM500		  ; 0.5 秒タイマ

	 MOVLW 	 B'00000100'	 ; ●○●●　リテラル → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ　PORTB に出力

	 CALL	 TIM500		  ; 0.5 秒タイマ

	 MOVLW 	 B'00001000'	 ; ○●●●　リテラル → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ　PORTB に出力

	 CALL	 TIM500		  ; 0.5 秒タイマ
	 GOTO	 MAIN		  ; MAINラベルへジャンプ

解答例集

;******************************
;	   0.4msec タイマ
;******************************

TIM04
	 MOVLW		  D'250'	
	 MOVWF		  T1
TIMLOOP1
	 NOP
	 DECFSZ		  T1, F
	 GOTO		  TIMLOOP1	 ; 1+1+(249 × 4)+3=1001 サイクル
	 RETURN				    ; 1001+2=1003 サイクル
	
;******************************
;	   100msec タイマ
;******************************

TIM100
	 MOVLW		  D'250'	
	 MOVWF		  T2
TIMLOOP2
	 CALL		  TIM04
	 DECFSZ		  T2, F
	 GOTO		  TIMLOOP2	 ; ≒ 0.4msec × 250=100msec
	 RETURN

;******************************
;	   500msec タイマ
;******************************

TIM500
	 MOVLW		  D'5'	
	 MOVWF		  T3
TIMLOOP3
	 CALL		  TIM100
	 DECFSZ		  T3, F
	 GOTO		  TIMLOOP3	 ; ≒ 100msec × 5=500msec
	 RETURN

END

解答例集

解答例集

20-1- 解答例

解答例集

STEP20-2.asm20-2- 解答例

〜 省略 〜　初期設定

;******************************
; 	  レジスタ設定
;******************************

	 T1	 EQU	 0CH
	 T2	 EQU	 0DH
	 T3	 EQU	 0EH

	 BSF	 STATUS, RP0	 ; BANK 1 に切替え
	 CLRF	 TRISA		  ; A ポートを全て出力に設定
	 CLRF	 TRISB		  ; B ポートを全て出力に設定
	 BCF	 STATUS, RP0	 ; BANK 0 に切替え

;******************************
; 　　　　メインリスト
;******************************

PRESET
	 MOVLW 	 B'00000001'	 ; 点灯 リテラル → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力

	 CALL	 TIM500		  ; 0.5 秒タイマ

MAIN
	 BCF 	 STATUS, 0	 ; STATUS の 0 ビット目 (Catrry) を 0 をセット
	 RLF	 PORTB, F	 ; PORTB の内容を左に 1ビットシフト

	 CALL	 TIM500		  ; 0.5 秒タイマ

	 BTFSC	 PORTB, 3	 ; RB3 が 0 ならば次をスキップ
	 GOTO	 PRESET		  ; (RB3 ≠ 0) ラベル PRESET にジャンプ
	 GOTO	 MAIN		  ; (RB3 ＝ 0) 無限ループ		

〜 省略 〜　タイマサブルーチン

END

電源 ON 後や，MCLR リセット後は
キャリービットの値は不定だから

STEP20-3.asm20-3- 解答例

〜 省略 〜　初期設定

;******************************
; 	  レジスタ設定
;******************************

	 T1	 EQU	 0CH
	 T2	 EQU	 0DH
	 T3	 EQU	 0EH

	 BSF	 STATUS, RP0	 ; BANK 1 に切替え
	 CLRF	 TRISA		  ; A ポートを全て出力に設定
	 CLRF	 TRISB		  ; B ポートを全て出力に設定
	 BCF	 STATUS, RP0	 ; BANK 0 に切替え

;******************************
; 　　　　メインリスト
;******************************

PRESET
	 MOVLW 	 B'00001000'	 ; 点灯 リテラル → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力

	 CALL	 TIM500		  ; 0.5 秒タイマ

MAIN
	 BCF 	 STATUS, 0	 ; STATUS の 0 ビット目 (Catrry) を 0 をセット
	 RRF	 PORTB, F	 ; PORTB の内容を左に 1ビットシフト

	 CALL	 TIM500		  ; 0.5 秒タイマ

	 BTFSC	 PORTB, 0	 ; RB0 が 0 ならば次をスキップ
	 GOTO	 PRESET		  ; (RB0 ≠ 0) ラベル PRESET にジャンプ
	 GOTO	 MAIN		  ; (RB0 ＝ 0) 無限ループ		

〜 省略 〜　タイマサブルーチン

END

RLF と違うのは赤文字部分だけ

解答例集

解答例集

STEP21.asm21- 解答例

〜 省略 〜　初期設定

;******************************

; 	  レジスタ設定

;******************************

	 T1	 EQU	 0CH

	 T2	 EQU	 0DH

	 T3	 EQU	 0EH

	 LED	 EQU	 0FH

	 BSF	 STATUS, RP0	 ; BANK 1 に切替え

	 CLRF	 TRISA		  ; A ポートを全て出力に設定

	 CLRF	 TRISB		  ; B ポートを全て出力に設定

	 BCF	 STATUS, RP0	 ; BANK 0 に切替え

	 CLRF	 LED		  ; 全ビット　0 → LED（カウンタ）

;******************************

; 　　　　メインリスト

;******************************

MAIN

	 CALL	 LED_PTN		 ; LED 点灯パターンテーブル参照

	 MOVWF 	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力

	 CALL	 TIM500		  ; 0.5 秒タイマ

	 INCF	 LED, F		  ; LED ＋１→ LED

	 MOVLW	 D'8'		  ; 8 → W レジスタ

	 SUBWF 	 LED,W		  ; LED − W レジスタ → W レジスタ

	 BTFSC	 STATUS,C	 ; キャリービットが 0ならば次をスキップ

	 CLRF	 LED		  ; 全ビット　0 → LED

	 GOTO	 MAIN		  ; 無限ループ

;******************************

;　　LED 点灯パターンテーブル

;******************************

LED_PTN

	 MOVF 	 LED, W		  ; LED → W レジスタ

	 ADDWF	 PCL, F		  ; W レジスタ＋ PCL レジスタ→ PCL レジスタ

	 RETLW	 B'00000000'	 ; ●●●● ●●●●

	 RETLW	 B'00000001'	 ; ●●●● ●●●○

	 RETLW	 B'00000011'	 ; ●●●● ●●○○

	 RETLW	 B'00000111'	 ; ●●●● ●○○○

	 RETLW	 B'00001111'	 ; ●●●● ○○○○

	 RETLW	 B'00001110'	 ; ●●●● ○○○●

	 RETLW	 B'00001100'	 ; ●●●● ○○●●

	 RETLW	 B'00001000'	 ; ●●●● ○●●●

〜 省略 〜　タイマサブルーチン

END

解答例集

解答例集

22-1- 解答例

シリーズ No.5　PIC 入門アセンブラ編

解答例集

STEP22-2.asm22-2- 解答例

〜 省略 〜　タイマサブルーチンで使う変数の設定と RA0 を入力に設定するのを忘れずに

;******************************
; 　　　　メインリスト
;******************************

MAIN
	 BTFSS 	 PORTA,0		 ; RA0 が 1 なら次の行をスキップ
	 GOTO	 LED_ONOFF	 ; (SW0 ON)
	 CALL 	 LED_OFF		 ; (SW0 OFF)
	 GOTO	 MAIN		  ; ループ

LED_ONOFF
	 CALL	 LED_ON		  ; LED 点灯処理へ
	 CALL	 TIM500		  ; 0.5 秒タイマ
	 CALL 	 LED_OFF		 ; LED 消灯処理へ
	 CALL	 TIM500		  ; 0.5 秒タイマ

	 GOTO	 MAIN		  ; ループ

;******************************
;	 LED 点灯処理
;******************************

LED_ON
	 MOVLW	 B'11111111'	 ; 点灯 リテラル → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → PORTB
	 RETURN

;******************************
;	 LED 消灯処理
;******************************

LED_OFF
	 MOVLW	 B'00000000'	 ; 消灯 リテラル → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → PORTB
	 RETURN

〜 省略 〜　タイマサブルーチン

END



21

シリーズ No.5　PIC 入門アセンブラ編

解答例集

STEP22-3.asm22-3- 解答例

〜 省略 〜　タイマサブルーチンで使う変数の設定と RA0 を入力に設定するのを忘れずに

	 CLRF	 PORTB		  ; B ポートの全ビットに 0をセット

;******************************

; 　　　　メインリスト

;******************************

MAIN

	 BTFSS 	 PORTA,0		 ; RA0 が 1 なら次の行をスキップ

	 GOTO	 LED_ONOFF	 ; (SW0 ON)

	 CALL 	 LED_OFF		 ; (SW0 OFF)

	 GOTO	 MAIN		  ; ループ

;******************************

;	 LED 点滅処理

;******************************

LED_ONOFF

	 MOVLW	 B'11111111'	 ; 反転ビット設定　リテラル → W レジスタ

	 XORWF	 PORTB,F		 ; PORTB を W レジスタで反転 → PORTB

	 CALL	 TIM500		  ; 0.5 秒タイマ

	 GOTO	 MAIN

;******************************

;	 LED 消灯処理

;******************************

LED_OFF

	 MOVLW	 B'00000000'	 ; 消灯 リテラル → W レジスタ

	 MOVWF	 PORTB		  ; W レジスタ → PORTB

	 RETURN

〜 省略 〜　タイマサブルーチン

END

電源 ON 後や，MCLR リセット後は
PORT A , PORT B のビット値は不定だから

「COMF」を使うと赤文字の２行を
COMF　PORTB,F
と書き換えます。

STEP23.asm23- 解答例

〜 省略 〜　CNT1,CNT2 変数の設定やタイマサブルーチンで使う変数の設定，
　　　　　　RA0,RA1 を入力に設定するのを忘れずに

;******************************
;  カウンタ値　初期設定
;******************************

INIT
	 MOVLW	 D'255'		  ; 255 → W レジスタ
	 MOVWF	 CNT1		  ; W レジスタ → CNT1
	 MOVLW	 D'127'		  ; 127 → W レジスタ
	 MOVWF	 CNT2		  ; W レジスタ → CNT2

;******************************
;  スイッチ入力 (SW0 ON?)
;******************************

MAIN
	 BTFSS 	 PORTA, 0	 ; RA0 が 1 なら次の行をスキップ
	 GOTO	 CASE1		  ; (SW0 ON) 
	 BTFSS 	 PORTA, 1	 ; RA1 が 1 なら次の行をスキップ
	 GOTO	 CASE2		  ; (SW1 ON) 
	 GOTO	 CASE3		  ; (SW0 OFF) (SW1 OFF)

;**********************************
; LED 点滅　状態１ (SW0 ON) 
;**********************************

CASE1
	 BTFSC	 PORTB,1		 ; LED1 が点灯していたら
	 BCF	 PORTB,1		 ; 消灯
	 CALL	 TIM04
	 DECFSZ	 CNT1,F		  ; CNT1 − 1 → CNT1　結果が 0なら次の行をスキップ
	 GOTO	 MAIN		  ;
	 MOVLW 	 B'00000001'   	 ; XORWF 用の反転ビット 1 のところだけ反転する
	 XORWF	 PORTB, F	 ; 点灯データを反転 (消灯←→点灯 )
	 MOVLW	 D'255'		  ; 255 → W レジスタ
	 MOVWF	 CNT1		  ; W レジスタ → CNT1
	 GOTO	 MAIN		  ;

解答例集

基本は黒文字部分のプログラムで正解ですが，SW1 を押したままで SW0 を押した時，LED が点灯しっぱなしにな
るのを防ぐには，青文字のプログラムを挿入するといいでしょう。

;**********************************
; LED 点滅　状態２ (SW1 ON)
;**********************************

CASE2
	 BTFSC	 PORTB,0		 ; LED0 が点灯していたら
	 BCF	 PORTB,0		 ; 消灯
	 CALL	 TIM04
	 DECFSZ	 CNT2,F		  ; CNT2 − 1 → CNT2　結果が 0なら次の行をスキップ
	 GOTO	 MAIN		  ;
	 MOVLW 	 B'00000010'   	 ; XORWF 用の反転ビット 1 のところだけ反転する
	 XORWF	 PORTB, F	 ; 点灯データを反転 (消灯←→点灯 )
	 MOVLW	 D'127'		  ; 127 → W レジスタ
	 MOVWF	 CNT2		  ; W レジスタ → CNT2
	 GOTO	 MAIN		  ;

;**********************************
; LED 消灯　状態３ (SW0 OFF) (SW1 OFF)
;**********************************

CASE3
	 MOVLW	 B'00000000'	 ; LED 消灯データ設定
	 MOVWF	 PORTB		  ; LED 消灯
	 GOTO	 INIT		  ;

〜 省略 〜　タイマサブルーチン

END

解答例集

解答例集

点滅状態をサブルーチン化するには，赤文字の部分を変更，追加します。

STEP23.asm23- 解答例

〜 省略 〜　CNT1,CNT2 変数の設定やタイマサブルーチンで使う変数の設定，
　　　　　　RA0,RA1 を入力に設定するのを忘れずに

;******************************
;  カウンタ値　初期設定
;******************************

INIT
	 MOVLW	 D'255'		  ; 255 → W レジスタ
	 MOVWF	 CNT1		  ; W レジスタ → CNT1
	 MOVLW	 D'127'		  ; 127 → W レジスタ
	 MOVWF	 CNT2		  ; W レジスタ → CNT2

;******************************
;  スイッチ入力 (SW0 ON?)
;******************************

MAIN
	 BTFSS 	 PORTA, 0	 ; RA0 が 1 なら次の行をスキップ
	 GOTO	 PRE_CASE1	 ; (SW0 ON) 
	 BTFSS 	 PORTA, 1	 ; RA1 が 1 なら次の行をスキップ
	 GOTO	 PRE_CASE2	 ; (SW1 ON) 
	 GOTO	 PRE_CASE3	 ; (SW0 OFF) (SW1 OFF)

;******************************
; CASE1 〜 3 を CALL RETURN で使うため
;******************************

PRE_CASE1
	 CALL CASE1
	 GOTO MAIN

PRE_CASE2
	 CALL CASE2
	 GOTO MAIN

PRE_CASE3
	 CALL CASE3
	 GOTO INIT

サブルーチン版

解答例集

;**********************************
; LED 点滅　状態１ (SW0 ON) 
;**********************************

CASE1
	 BTFSC	 PORTB,1		 ; LED1 が点灯していたら
	 BCF	 PORTB,1		 ; 消灯
	 CALL	 TIM04
	 DECFSZ	 CNT1,F		  ; CNT1 − 1 → CNT1　結果が 0なら次の行をスキップ
	 GOTO	 MAIN		  ;
	 MOVLW 	 B'00000001'   	 ; XORWF 用の反転ビット 1 のところだけ反転する
	 XORWF	 PORTB, F	 ; 点灯データを反転 (消灯←→点灯 )
	 MOVLW	 D'255'		  ; 255 → W レジスタ
	 MOVWF	 CNT1		  ; W レジスタ → CNT1
	 RETURN

;**********************************
; LED 点滅　状態２ (SW1 ON)
;**********************************

CASE2
	 BTFSC	 PORTB,0		 ; LED0 が点灯していたら
	 BCF	 PORTB,0		 ; 消灯
	 CALL	 TIM04
	 DECFSZ	 CNT2,F		  ; CNT2 − 1 → CNT2　結果が 0なら次の行をスキップ
	 GOTO	 MAIN		  ;
	 MOVLW 	 B'00000010'   	 ; XORWF 用の反転ビット 1 のところだけ反転する
	 XORWF	 PORTB, F	 ; 点灯データを反転 (消灯←→点灯 )
	 MOVLW	 D'127'		  ; 127 → W レジスタ
	 MOVWF	 CNT2		  ; W レジスタ → CNT2
	 RETURN

;**********************************
; LED 消灯　状態３ (SW0 OFF) (SW1 OFF)
;**********************************

CASE3
	 MOVLW	 B'00000000'	 ; LED 消灯データ設定
	 MOVWF	 PORTB		  ; LED 消灯
	 RETURN

〜 省略 〜　タイマサブルーチン

END

STEP24.asm24- 解答例

〜 省略 〜　CNT1,CNT2,CNT3 変数の設定やタイマサブルーチンで使う変数の設定，
　　　　　　RA0,RA1 を入力に設定するのを忘れずに

;******************************
;  カウンタ値　初期設定
;******************************

INIT
	 MOVLW	 D'255'		  ; 255 → W レジスタ
	 MOVWF	 CNT1		  ; W レジスタ → CNT1
	 MOVLW	 D'127'		  ; 127 → W レジスタ
	 MOVWF	 CNT2		  ; W レジスタ → CNT2
	 MOVLW	 D'63'		  ; 63 → W レジスタ
	 MOVWF	 CNT3		  ; W レジスタ → CNT3

;******************************
;  スイッチ入力 (SW0 ON?)
;******************************

MAIN
	 BTFSS 	 PORTA, 0		 ; RA0 が 1 なら次の行をスキップ
	 GOTO	 CK_SW_1		  ; (SW0 ON)
	 GOTO	 CK_SW_2		  ; (SW0 OFF)

;******************************
;  スイッチ入力 (SW1 ON?)
;******************************

CK_SW_1
	 BTFSS 	 PORTA, 1		 ; RA1 が 1 なら次の行をスキップ
	 GOTO 	 CASE4		  ; (SW0 ON) (SW1 ON)
	 GOTO 	 CASE1		  ; (SW0 ON) (SW1 OFF)

;******************************
;  スイッチ入力 (SW1 ON?)
;******************************

CK_SW_2
	 BTFSS 	 PORTA, 1		 ; RA1 が 1 なら次の行をスキップ
	 GOTO 	 CASE2		  ; (SW0 OFF) (SW1 ON)
	 GOTO 	 CASE3		  ; (SW0 OFF) (SW1 OFF)

解答例集

基本は黒文字部分のプログラムで正解ですが，SW を押すタイミングによって同時に点灯しなかったり，点灯しっぱ
なしになるのを防ぐには青文字のプログラムを挿入するといいでしょう。

;**********************************
; LED 点滅　状態１ (SW0 ON) (SW1 OFF)
;**********************************

CASE1
	 BTFSC	 PORTB,1		  ; LED1 が点灯していたら
	 BCF	 PORTB,1		  ; 消灯
	 CALL	 TIM04
	 DECFSZ	 CNT1,F		  ; CNT1 − 1 → CNT1　結果が 0なら次の行をスキップ
	 GOTO	 MAIN		  ;
	 MOVLW 	 B'00000001'   	 ; XORWF 用の反転ビット 1 のところだけ反転する
	 XORWF	 PORTB, F		 ; 点灯データを反転 (消灯←→点灯 )
	 MOVLW	 D'255'		  ; 255 → W レジスタ
	 MOVWF	 CNT1		  ; W レジスタ → CNT1
	 GOTO	 MAIN		  ;

;**********************************
; LED 点滅　状態２ (SW0 OFF) (SW1 ON)
;**********************************

CASE2
	 BTFSC	 PORTB,0		  ; LED0 が点灯していたら
	 BCF	 PORTB,0		  ; 消灯
	 CALL	 TIM04
	 DECFSZ	 CNT2,F		  ; CNT2 − 1 → CNT2　結果が 0なら次の行をスキップ
	 GOTO	 MAIN		  ;
	 MOVLW 	 B'00000010'   	 ; XORWF 用の反転ビット 1 のところだけ反転する
	 XORWF	 PORTB, F		 ; 点灯データを反転 (消灯←→点灯 )
	 MOVLW	 D'127'		  ; 127 → W レジスタ
	 MOVWF	 CNT2		  ; W レジスタ → CNT2
	 GOTO	 MAIN		  ;

;**********************************
; LED 消灯　状態３ (SW0 OFF) (SW1 OFF)
;**********************************

CASE3
	 MOVLW	 B'00000000'	 ; LED 消灯データ設定
	 MOVWF	 PORTB		  ; LED 消灯
	 GOTO	 INIT		  ;

;**********************************
; LED 点滅　状態４ (SW0 ON) (SW1 ON)
;**********************************

CASE4

解答例集

STEP25.asm25-1- 解答例

〜 省略 〜　初期設定

;******************************
; 	  レジスタ設定
;******************************

	 CNT	 EQU	 0CH

	 BSF	 STATUS, RP0	 ; BANK 1 に切替え
	 MOVLW	 B'00001'	 ; RA0 を入力に設定 → W レジスタ
	 MOVWF	 TRISA		  ; W レジスタ → A ポート 入出力設定
	 CLRF	 TRISB		  ; B ポートを全て出力に設定
	 BCF	 STATUS, RP0	 ; BANK 0 に切替え

	 CLRF	 CNT		  ; CNT をクリア
	 CLRF	 PORTB		  ; B ポートの全ビットに 0をセット

;******************************
;   　　スイッチ入力 (SW0 ON?)
;******************************
LOOP
	 BTFSS 	 PORTA, 0	 ; RA0 が 0 なら次の行をスキップ

解答例集

	 BTFSC	 PORTB,0		  ; LED0 が点灯していたら
	 BSF	 PORTB,1		  ; LED1 も点灯
	 BTFSC	 PORTB,1		  ; LED1 が点灯していたら
	 BSF	 PORTB,0		  ; LED0 も点灯
	 CALL	 TIM04
	 DECFSZ	 CNT3,F		  ; CNT3 − 1 → CNT3　結果が 0なら次の行をスキップ
	 GOTO	 MAIN		  ;
	 MOVLW 	 B'00000011' 	 ; XORWF 用の反転ビット 1 のところだけ反転する
	 XORWF	 PORTB, F		 ; 点灯データを反転 (消灯←→点灯 )
	 MOVLW	 D'63'		  ; 63 → W レジスタ
	 MOVWF	 CNT3		  ; W レジスタ → CNT3
	 GOTO	 MAIN		  ;

〜 省略 〜　タイマサブルーチン

END

解答例集

26-1- 解答例

	 GOTO	 LED_ON		  ; (SW0 ON)
	 GOTO	 LOOP		  ; (SW0 OFF)　ループ

;******************************
;	 LED2 進数表示処理 (SW0 ON)
;******************************

LED_ON
	 INCF	 CNT, F		  ; CNT ＋ 1 → CNT
	 MOVF	 CNT,W		  ; CNT → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → PORTB
	 GOTO	 LOOP		  ; スイッチ入力待ちループ

END

STEP26.asm26-2- 解答例

〜 省略 〜　CNT 変数の設定やタイマサブルーチンで使う変数の設定，

　　　　　　RA0 を入力に設定するのを忘れずに

	 CLRF	 CNT		  ; CNT をクリア

	 CLRF	 PORTB		  ; B ポートの全ビットに 0をセット

;******************************

;   　　スイッチ入力 (SW0 ON?)

;******************************

LOOP

	 BTFSS 	 PORTA, 0	 ; RA0 が 1 なら次の行をスキップ

	 GOTO	 LED_ON		  ; (SW0 ON)

	 GOTO	 LOOP		  ; (SW0 OFF)　ループ

;******************************

;	 LED ２進数表示処理 (SW0 ON)

;******************************

LED_ON

	 INCF	 CNT, F		  ; CNT ＋ 1 → CNT

	 MOVF	 CNT,W		  ; CNT → W レジスタ

	 MOVWF	 PORTB		  ; W レジスタ → PORTB

	 CALL	 TIM100		  ; 100msec タイマ

	 GOTO	 LOOP		  ; スイッチ入力待ちループ

〜 省略 〜　タイマサブルーチン

END

解答例集

解答例集

STEP27-1.asm27-1- 解答例

〜 省略 〜　CNT 変数の設定，RA0 を入力に設定するのを忘れずに

	 CLRF	 CNT		  ; CNT をクリア

	 CLRF	 PORTB		  ; B ポートの全ビットに 0をセット

;******************************

;   　　スイッチ入力 (SW0 ON?)

;******************************

LOOP1

	 BTFSC 	 PORTA, 0	 ; RA0 が 0 なら次の行をスキップ

	 GOTO	 LOOP1		  ; (SW0 OFF)　スイッチ入力待ちループ

;********************************

; スイッチ開放チェック &カウント値加算 (SW0 ON)

;********************************

LOOP2

	 BTFSS	 PORTA, 0	 ; RA0 が 1 なら次の行をスキップ

	 GOTO	 LOOP2		  ; (SW0 ON)　スイッチ開放待ちループ

	 INCF	 CNT, F		  ; (SW0 OFF) CNT ＋ 1 → CNT

	

;******************************

;	 LED2 進数表示処理 (SW0 OFF)

;******************************

	 MOVF	 CNT, W		  ; CNT → W レジスタ

	 MOVWF	 PORTB		  ; W レジスタ → PORTB

	 GOTO 	 LOOP1		  ; スイッチ入力待ちループ

END

「STEP27-2.asm」は，ここに

CALL　TIM100

を入れて，タイマサブルーチンつけるだけ。

変数の設定も忘れずに。

STEP28.asm28- 解答例

〜 省略 〜　初期設定

;******************************
; 	  レジスタ設定
;******************************

	 T1	 EQU	 0CH
	 T2	 EQU	 0DH

	 BSF	 STATUS, RP0	 ; BANK 1 に切替え
	 MOVLW	 B'00001'	 ; RA0 を入力に設定 → W レジスタ
	 MOVWF	 TRISA		  ; W レジスタ → A ポート 入出力設定
	 CLRF	 TRISB		  ; B ポートを全て出力に設定
	 BCF	 STATUS, RP0	 ; BANK 0 に切替え

	 CLRF	 PORTB		  ; B ポートの全ビットに 0をセット

;******************************
;  　スイッチ入力 (SW0 ON?)
;******************************

MAIN
	 BTFSS 	 PORTA, 0	 ; RA0 が 1 なら次の行をスキップ
	 CALL	 SPK_ON		  ; (SW0 ON)
	 CALL	 SPK_OFF		 ; (SW0 OFF)
	 GOTO	 MAIN		  ; ループ
	
;******************************
;   スピーカ ON の処理 (SW0 ON)
;******************************

SPK_ON
	 CALL	 TIM2		  ; 2msec(500Hz) タイマ
	 MOVLW	 B'11111111'	 ; リテラル → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → B ポートに出力
	 CALL	 TIM2		  ; 2msec(500Hz) タイマ
	 RETURN			 

STEP22 と全く同じでは面白くないので，少し変えてあります。
「SPK_ON」ラベルへジャンプした後の「RETURN」の戻り先がミソです。

解答例集

;******************************
;     スピーカ OFF の処理 (SW0 OFF)
;******************************

SPK_OFF
	 MOVLW	 B'00000000'	 ; リテラル → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → B ポートに出力
	 RETURN
	
;******************************
;	   0.4msec タイマ
;******************************

TIM04
	 MOVLW	 D'250'
	 MOVWF	 T1
TIMLOOP1
	 NOP
	 DECFSZ	 T1, F
	 GOTO	 TIMLOOP1	 ; 1+1+(249 × 4)+3=1001 サイクル
	 RETURN	 		  ; 1001+2=1003 サイクル

;******************************
;	   2msec タイマ
;******************************

TIM2
	 MOVLW	 D'5'	
	 MOVWF	 T2
TIMLOOP2
	 CALL	 TIM04
	 DECFSZ	 T2, F
	 GOTO	 TIMLOOP2	 ; ≒ 0.4msec × 5=2msec
	 RETURN

END

全 B ポートを同じように出力しています。LED が接続されていれば点滅しますが．．．速すぎて見えないですね。

解答例集

STEP28.asm28- 解答例

〜 省略 〜　初期設定

;******************************
; 	  レジスタ設定
;******************************

	 T1	 EQU	 0CH
	 BEEP_C	 EQU	 0DH

	 BSF	 STATUS, RP0	 ; BANK 1 に切替え
	 MOVLW	 B'00001'	 ; RA0 を入力に設定 → W レジスタ
	 MOVWF	 TRISA		  ; W レジスタ → A ポート 入出力設定
	 CLRF	 TRISB		  ; B ポートを全て出力に設定
	 BCF	 STATUS, RP0	 ; BANK 0 に切替え

	 CLRF	 PORTB		  ; B ポートの全ビットに 0をセット
	 CALL	 INT_BEEP_C	 ; BEEP_C の初期化

;******************************
;   スイッチ入力 (SW0 ON?)
;******************************

MAIN
	 BTFSS 	 PORTA, 0	 ; RA0 が 1 なら次の行をスキップ
	 GOTO	 BEEP_LOOP	 ; (SW0 ON)
	 CLRF	 PORTB		  ; (SW0 OFF)
	 GOTO	 MAIN		  ; ループ

;******************************
;     スピーカ ON の処理 (SW0 ON)
;******************************

BEEP_LOOP
	 CALL	 TIM01		  ; 0.1msec TIMER
	 DECFSZ	 BEEP_C, F	 ; BEEP_C − 1 → BEEP_C　結果が 0なら次の行をスキップ
	 GOTO	 MAIN		  ; (BEEP_C ≠ 0)
	 GOTO	 BEEP_RET	 ; (BEEP_C ＝ 0)

周波数設定カウンタと「COMF」命令を使ったプログラムです。
周波数設定カウンタ BEEP_C の値を変更するだけで，簡単に周波数を変えることができます。

解答例集

;******************************
;    スピーカ状態の反転の処理
;******************************

BEEP_RET
	 COMF	 PORTB, F	 ; B ポートの内容を反転
	 CALL	 INT_BEEP_C	 ; BEEP_C の初期化
	 GOTO	 MAIN		  ;

;******************************
;    BEEP_C の初期化
;******************************

INT_BEEP_C
	 MOVLW	 D'20'		  ;BEEP_C の値を初期化
	 MOVWF	 BEEP_C
	 RETURN
	
;******************************
;	   0.1msec タイマ
;******************************

TIM01
	 MOVLW	 D'61'	
	 MOVWF	 T1
TIMLOOP1
	 NOP
	 DECFSZ	 T1, F
	 GOTO	 TIMLOOP1	 ; 1+1+(61 × 4)+3=249 サイクル
	 RETURN			   ; 249+2=251 サイクル

END

解答例集

0.1msec タイマを 20 回繰り返して

2msec としています。

BEEP_C は 1 〜 255 までの値を設定できるので，0.1msec 〜 25.5msec 周期の音を出すことができます。
周波数でいうと，10000Hz 〜 39.2Hz ですね。



36

シリーズ No.5　PIC 入門アセンブラ編

RB4 ポートのみ出力しています。

STEP29.asm29- 解答例

〜 省略 〜　CNT1,CNT2,CNT3 変数の設定やタイマサブルーチンで使う変数の設定，
　　　　　　RA0,RA1 を入力に設定するのを忘れずに

;******************************
;  カウンタ値　初期設定
;******************************

	 MOVLW	 D'8'		  ; 8 → W レジスタ
	 MOVWF	 CNT1		  ; W レジスタ → CNT1
	 MOVLW	 D'4'		  ; 4 → W レジスタ
	 MOVWF	 CNT2		  ; W レジスタ → CNT2
	 MOVLW	 D'2'		  ; 2 → W レジスタ
	 MOVWF	 CNT3		  ; W レジスタ → CNT3

;******************************
;  スイッチ入力 (SW0 ON?)
;******************************

MAIN
	 BTFSS 	 PORTA, 0	 ; RA0 が 1 なら次の行をスキップ
	 GOTO	 CK_SW_1		 ; (SW0 ON)
	 GOTO	 CK_SW_2		 ; (SW0 OFF)

;******************************
;  スイッチ入力 (SW1 ON?)
;******************************

CK_SW_1
	 BTFSS 	 PORTA, 1	 ; RA1 が 1 なら次の行をスキップ
	 GOTO 	 CASE4		  ; (SW0 ON) (SW1 ON)
	 GOTO 	 CASE1		  ; (SW0 ON) (SW1 OFF)

;******************************
;  スイッチ入力 (SW1 ON?)
;******************************

CK_SW_2
	 BTFSS 	 PORTA, 1	 ; RA1 が 1 なら次の行をスキップ
	 GOTO 	 CASE2		  ; (SW0 OFF) (SW1 ON)
	 GOTO 	 CASE3		  ; (SW0 OFF) (SW1 OFF)

;**********************************
; スピーカ 312.5Hz (SW0 ON) (SW1 OFF)
;**********************************

CASE1
	 CALL	 TIM04
	 DECFSZ	 CNT1,F		  ; CNT1 − 1 → CNT1　結果が 0なら次の行をスキップ

解答例集

	 GOTO	 MAIN		  ; (CNT1 ≠ 0)
	 CALL	 SPK_ON		  ; (CNT1 ＝ 0)
	 MOVLW	 D'8'		  ; 8 → W レジスタ
	 MOVWF	 CNT1		  ; W レジスタ → CNT1
	 GOTO	 MAIN		  ;

;**********************************
; スピーカ 625Hz (SW0 OFF) (SW1 ON)
;**********************************

CASE2
	 CALL	 TIM04
	 DECFSZ	 CNT2,F		  ; CNT2 − 1 → CNT2　結果が 0なら次の行をスキップ
	 GOTO	 MAIN		  ; (CNT2 ≠ 0)
	 CALL	 SPK_ON		  ; (CNT2 ＝ 0)
	 MOVLW	 D'4'		  ; 4 → W レジスタ
	 MOVWF	 CNT2		  ; W レジスタ → CNT2
	 GOTO	 MAIN		  ;

;**********************************
; スピーカ オフ (SW0 OFF) (SW1 OFF)
;**********************************

CASE3
	 MOVLW	 B'00000000'	 ; LED 消灯データ設定
	 MOVWF	 PORTB		  ; LED 消灯
	 GOTO	 MAIN		  ;

;**********************************
; L スピーカ 1250Hz (SW0 ON) (SW1 ON)
;**********************************

CASE4
	 CALL	 TIM04
	 DECFSZ	 CNT3,F		  ; CNT3 − 1 → CNT3　結果が 0なら次の行をスキップ
	 GOTO	 MAIN		  ; (CNT3 ≠ 0)
	 CALL	 SPK_ON		  ; (CNT3 ＝ 0)
	 MOVLW	 D'2'		  ; 2 → W レジスタ
	 MOVWF	 CNT3		  ; W レジスタ → CNT3
	 GOTO	 MAIN		  ;

;**********************************
; スピーカ　オン
;**********************************

SPK_ON
	 MOVLW 	 B'00010000' 	 ; XORWF 用の反転ビット 1 のところだけ反転する
	 XORWF	 PORTB, F	 ; データを反転 (ON ←→ OFF)
	 RETURN

〜 省略 〜　タイマサブルーチン

END

解答例集

STEP30.asm30- 解答例

〜 省略 〜　初期設定

;******************************
; 	 　レジスタ設定
;******************************

	 BSF	 STATUS, RP0	 ; BANK 1 に切替え
	 MOVLW 	 B'00001'	 ; RA0 を入力に設定 → W レジスタ
	 MOVWF	 TRISA		  ; W レジスタ → A ポート　入出力設定
	 CLRF	 TRISB		  ; B ポートを全て出力に設定
	 BCF	 STATUS, RP0	 ; BANK 0 に切替え

;******************************
; 　　　　メインリスト
;******************************

MAIN
	 BTFSC	 PORTA,0		 ; RA0 が 0 なら次の行をスキップ
	 GOTO	 MOT_OFF		 ; MOT_OFF へジャンプ（SW0 OFF）
	 GOTO	 MOT_ON		  ; MOT_ON へジャンプ（SW0 ON）

;******************************
;	 モータ回転処理
;******************************

MOT_ON
	 MOVLW	 B'00010000'	 ; RB4 ON → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力
	 GOTO	 MAIN		  ; MAIN へジャンプ

;******************************
;	 モータ停止処理
;******************************

MOT_OFF
	 MOVLW	 B'00000000'	 ; RB4 OFF → W レジスタ
	 MOVWF	 PORTB		  ; W レジスタ → ファイルレジスタ PORTB に出力
	 GOTO	 MAIN		  ; MAIN へジャンプ

END

解答例集

解答例集

STEP31.asm31- 解答例

〜 省略 〜　初期設定

;******************************
; 	 　レジスタ設定
;******************************

	 T1	 EQU	 0CH
	 T2	 EQU	 0DH

	 BSF	 STATUS, RP0	 ; BANK 1 に切替え
	 MOVLW 	 B'00011'	 ; RA0,RA1 を入力に設定 → W レジスタ
	 MOVWF	 TRISA		  ; W レジスタ → A ポート　入出力設定
	 CLRF	 TRISB		  ; B ポートを全て出力に設定
	 BCF	 STATUS, RP0	 ; BANK 0 に切替え

;******************************
;  スイッチ入力 (スイッチ 0 ON?)
;******************************

MAIN
	 BTFSS 	 PORTA, 0	 ; RA0 が 1 なら次の行をスキップ
	 GOTO	 CK_SW1_CHK	 ; (SW0 ON)
	 CALL	 MOT_OFF		 ; (SW0 OFF) モータ停止処理へ
	 GOTO	 MAIN		  ; スイッチ入力チェックループ

;******************************
;  スイッチ入力 (スイッチ 1 ON?)
;******************************

CK_SW1_CHK
	 BTFSS 	 PORTA, 1	 ; RA1 が 1 なら次の行をスキップ
	 GOTO	 MOT_80		  ; (SW1 ON) DUTY 比 80% のモータ回転へ
	 GOTO	 MOT_20		  ; (SW1 OFF) DUTY 比 20% のモータ回転へ

;******************************
;  DUTY 比 80% のモータ回転の処理
;******************************

MOT_80
	 CALL	 MOT_ON

	 CALL	 TIM8		  ; 8msec タイマ
	 CALL	 MOT_OFF
	 CALL	 TIM2		  ; 2msec タイマ
	 GOTO	 MAIN
	
;******************************
;  DUTY 比 20% のモータ回転の処理
;******************************

MOT_20
	 CALL	 MOT_ON
	 CALL	 TIM2		  ; 2msec タイマ
	 CALL	 MOT_OFF
	 CALL	 TIM8		  ; 8msec タイマ
	 GOTO	 MAIN

;******************************
;	 モータ回転の処理
;******************************

MOT_ON
	 MOVLW	 B'00010000'	 ; B ポートの RB4 に 1 をセット
	 MOVWF	 PORTB		  ; B ポートに出力 (モータ回転 )
	 RETURN

;******************************
;	 モータ停止の処理
;******************************

MOT_OFF
	 MOVLW	 B'00000000'	 ; B ポートの全ビットに 0をセット
	 MOVWF	 PORTB		  ; B ポートに出力 (モータ停止 )
	 RETURN

;******************************
;	   0.4msec タイマ
;******************************

TIM04
	 MOVLW	 D'250'		  ; ループ回数 250 をセット

解答例集

解答例集

	 MOVWF	 T1		  ;
TIMLOOP1
	 NOP
	 DECFSZ	 T1, F
	 GOTO	 TIMLOOP1	 ; 1+1+(249 × 4)+3=1001 サイクル
	 RETURN			   ; 1001+2=1003 サイクル

;******************************
;	   2msec タイマ
;******************************

TIM2
	 MOVLW	 D'5'		  ; ループ回数 5をセット
	 MOVWF	 T2
TIMLOOP2
	 CALL	 TIM04
	 DECFSZ	 T2, F
	 GOTO	 TIMLOOP2	 ; ≒ 0.4msec × 5=2msec
	 RETURN

;******************************
;	   8msec タイマ
;******************************

TIM8
	 MOVLW	 D'20'		  ; ループ回数 20 をセット
	 MOVWF	 T2
TIMLOOP3
	 CALL	 TIM04
	 DECFSZ	 T2, F
	 GOTO	 TIMLOOP3	 ; ≒ 0.4msec × 20=8msec
	 RETURN

END

STEP32.asm32- 解答例

 list      p=16F84A
 #include <p16F84A.inc>

 __CONFIG _HS_OSC & _WDT_OFF & _PWRTE_ON & _CP_OFF

	 ORG	 00h	 	 ; 00h 番地に以下の命令をセット
	 GOTO	 INIT		  ; 初期化処理
	 ORG	 04h	 	 ; 04h 番地に以下の命令をセット
	 GOTO	 INTER		  ; 割込み処理

;******************************
; 	 初期化処理
;******************************

INIT	
	 T1		  EQU	 0CH	
	 T2    		  EQU	 0DH	
	 START_FLG	 EQU	 0EH	 ; スタートフラグ
	 WREG_BU		 EQU	 0FH	 ; W レジスタ退避用	
	 STATUS_BU	 EQU	 10H	 ; STATUS レジスタ退避用	

	 BSF   	 STATUS,RP0	 ; BANK 1 に切替え
	 MOVLW 	 B'00001' 	 ; RA0 を入力に設定
	 MOVWF 	 TRISA	 	 ; W レジスタ→ Aポート　入出力設定
	 MOVLW 	 B'00000001' 	 ; RB0 を入力に設定→ Wレジスタ
	 MOVWF 	 TRISB		  ; W レジスタ→ Bポート　入出力設定
	 MOVLW 	 B'10111111' 	 ; RB0/INT 割込み要因を立下りエッジに設定
	 MOVWF 	 OPTION_REG	 ; 
	 BCF   	 STATUS,RP0	 ; BANK 0 に切替え

	 CLRF	 PORTB		  ; B ポートの出力ビットに 0をセット
	 MOVLW 	 B'00000001' 	 ; 1 → START_FLG
	 MOVWF 	 START_FLG	 ;
	 BSF   	 INTCON,INTE	 ; 1 → INTE　RB0/INT 割込み許可
	 BSF   	 INTCON,GIE	 ; 1 → GIE　全体割込み許可

;******************************
; 　　　　メインリスト
;******************************

MAIN
	 BTFSS   PORTA,0		 ; RA0 が 1 なら次の行をスキップ

	 GOTO 	 SWA_ON	 	 ; (SW0 ON)
	 GOTO 	 SWA_OFF		 ; (SW0 OFF)
	
SWA_ON
	 BTFSC   START_FLG,0	 ; START_FLG が 0 なら次の行をスキップ
	 GOTO 	 SFLG_1		  ; (START_FLG ≠ 0)
	 GOTO 	 SFLG_0		  ; (START_FLG ＝ 0)

SWA_OFF
	 MOVLW 	 B'00000001'	 ; 1 → START_FLG
	 MOVWF 	 START_FLG	 ;
	 BCF	 PORTB,4 	 ; 0 → RB4
	 GOTO 	 MAIN		  ; MAIN へ
	
SFLG_1
	 BSF	 PORTB,4 	 ; 1 → RB4
	 GOTO 	 MAIN		  ; MAIN へ
	
SFLG_0
	 BCF	 PORTB,4 	 ; 0 → RB4
	 GOTO 	 MAIN		  ; MAIN へ

;******************************
; 	 割込み処理
;******************************

INTER
				    ; 退避　Wレジスタ，STATUS レジスタ
	 MOVWF	 WREG_BU		 ; W レジスタ→ WREG_BU
	 SWAPF	 STATUS,W	 ; STATUS レジスタ→ STATUS_BU　上下入替
	 MOVWF	 STATUS_BU	 ; 

	 BCF	 INTCON,INTF	 ; 0 → INTF　RB0/INT 割込みフラグクリア
	 BCF	 PORTB,4 	 ; 0 → RB4
	 MOVLW 	 B'00000000' 	 ; 0 → START_FLG
	 MOVWF 	 START_FLG	 ;

				    ; 復帰　Wレジスタ，STATUS レジスタ
	 SWAPF	 STATUS_BU,W	 ; STATUS_BU → STATUS レジスタ　上下入替
	 MOVWF	 STATUS		  ;
	 SWAPF WREG_BU,F		 ; WREG_BU 上下入替
	 SWAPF WREG_BU,W		 ; WREG_BU → W レジスタ 上下入替

	 RETFIE			   ; 0 → GIE　全体割込み許可して元のルーチンへ

END

STEP33.asm33- 解答例

 list      p=16F84A
 #include <p16F84A.inc>

 __CONFIG _HS_OSC & _WDT_OFF & _PWRTE_ON & _CP_OFF

	 ORG	 00h	 	 ; 00h 番地に以下の命令をセット
	 GOTO	 INIT		  ; 初期化処理
	 ORG	 04h	 	 ; 04h 番地に以下の命令をセット
	 GOTO	 INTER		  ; 割込み処理

;******************************
; 	 初期化処理
;******************************

INIT
	 OVFL_C		  EQU	 0CH	 ; オーバーフロー回数カウンタ
	 WREG_BU		 EQU 	 0DH	 ; W レジスタ退避用
	 STATUS_BU	 EQU	 0EH	 ; STATUS レジスタ退避用

	 BSF	 STATUS,RP0	 ; BANK 1 に切替
	 MOVLW	 B'00001' 	 ; RA0 を入力に設定
	 MOVWF	 TRISA		  ; W レジスタ→ Aポート　入出力設定
	 CLRF	 TRISB		  ; B ポートを全て出力に設定
	 MOVLW 	 B'10000111'	 ; プリスケーラ 256 をセット
	 MOVWF 	 OPTION_REG	 ; W レジスタ→ OPTION レジスタ
	 BCF   	 STATUS,RP0	 ; BANK 0 に切替

	 CLRF	 PORTB		  ; B ポートの出力ビットに 0をセット
	 MOVLW	 D'12'		  ; 12 → TMR0
	 MOVWF	 TMR0		  ; 
	 MOVLW	 D'20'		  ; 20 → OVFL_C
	 MOVWF	 OVFL_C	 	 ;
	
;******************************
;  スイッチ入力 (スイッチ 1 ON?)
;******************************

MAIN
	 BTFSS	 PORTA,0		 ; RA0 が 0 なら次の行をスキップ
	 GOTO	 SW1_ON		  ; (SW0 ON)
	 GOTO	 SW1_OFF		 ; (SW0 OFF)

SW1_ON
	 BSF	 INTCON,T0IE	 ; 1 → T0IE　タイマ割り込み許可
	 BSF	 INTCON,GIE	 ; 1 → GIE　割り込み全般許可
	 GOTO	 MAIN		  ; MAIN へ

SW1_OFF
	 BCF	 INTCON,T0IE	 ; 0 → T0IE　値タイマ割り込み禁止
	 CALL 	 LED_OFF		 ; LED_OFF サブルーチン実行
	 GOTO	 MAIN		  ; MAIN へ

;******************************
;	 タイマ割込みルーチン
;******************************

INTER
				    ; 退避　Wレジスタ，STATUS レジスタ
	 MOVWF	 WREG_BU		 ; W レジスタ→ WREG_BU
	 SWAPF	 STATUS,W	 ; STATUS レジスタ→ STATUS_BU　上下入替
	 MOVWF	 STATUS_BU	 ; 

	 BCF	 INTCON,T0IF	 ; 0 → T0IF　TMR0 のフラグクリア
	 MOVLW 	 D'12'		  ; 12 → TMR0
	 MOVWF 	 TMR0		  ;
	 DECFSZ	 OVFL_C,F	 ; OVFL_C − 1 → OVFL_C　OVFL_C が 0 なら次の行をスキップ
	 GOTO	 RET_INTER	 ;（OVFL_C ≠ 0）
	 MOVLW 	 D'20'		  ; （OVFL_C ＝ 0）20 → OVFL_C 
	 MOVWF 	 OVFL_C	 	 ; 
	
	 CALL    LED_ONOFF	 ; LED_ONOFF サブルーチン

RET_INTER
	 			   ; 復帰　Wレジスタ，STATUS レジスタ
	 SWAPF	 STATUS_BU,W	 ; STATUS_BU → STATUS レジスタ　上下入替
	 MOVWF	 STATUS		  ;
	 SWAPF	 WREG_BU,F	 ; WREG_BU 上下入替
	 SWAPF	 WREG_BU,W	 ; WREG_BU → W レジスタ 上下入替

	 RETFIE			   ; 0 → GIE　全体割込み許可して元のルーチンへ

;******************************
;	 LED 点滅処理
;******************************

LED_ONOFF
	 MOVLW 	 B'00000001' 	 ; RB0 を反転ビット指定
	 XORWF 	 PORTB,F		 ; B ポートに反転出力 (LED 点滅 )
	 RETURN	 		  ;

;******************************
;	 LED 消灯処理
;******************************

LED_OFF
	 MOVLW 	 B'00000000' 	 ; B ポートの全ビットに 0をセット
	 MOVWF 	 PORTB		  ; B ポートに出力 (LED 消灯 )
	 RETURN			   ;

END




